

Clojure/conj 2015 test.check Workshop

Gary Fredericks

November 15, 2015

- Property-based testing
 - an approach to testing software
- `test.check`
 - a clojure library for writing property-based tests

- Gary Fredericks (@gfredericks_)
- Using Clojure professionally for ~6 years
 - Currently at Groupon
- User/Maintainer of `test.check`

Property-Based Testing

- Main idea
 - Describe possible "inputs" to your program
 - Describe properties that hold for all inputs
 - Testing framework generates random inputs, checks the properties
- Benefits
 - Get much better test coverage than example-based tests, especially when testing combinations of features & edge-cases
 - Find bugs in tools/libraries
- History
 - Originated in Haskell-land as "QuickCheck"

```
1 (def gen-record
2   ...)
3
4 (defspec db-roundtrip 1000
5   (prop/for-all [record gen-record]
6     (do
7       (save-to-db record)
8       (= record (read-from-db (:id record)))))))
```

Bugs I've Found

- Postgres serializable transactions (BUG #13667)
 - Didn't work correctly for some write-before-read uses
- Google Closure integers
 - Base 36 serialization
 - Dividing big-integers larger than Double/MAX_VALUE
- Carmine message queues
 - Inadvertently required two connections per worker, allowing deadlock while trying to acquire second connection

- A QuickCheck-inspired testing library in Clojure
- Created by Reid Draper (originally called "simple-check")
- Integrates with `clojure.test`
- Supports ClojureScript

Goals for this Workshop

- How do I use this thing?
 - Writing generators
 - Writing properties
- What tests should I write?

Schedule

- I talk about generators
- You write some generators
- I talk about properties & shrinking
- You write some properties
- I talk about property-based testing in real life
- You go back to your job and do real-life property-based testing

Gentlefolks, start your repls

Visit & clone `http://bit.ly/1MvJH3D`

OR

```
git clone \  
https://github.com/gfredericks/test-check-workshop.git
```

AND THEN

```
cd test-check-workshop/part-2  
lein repl
```

Generators

Generators In Action

```
1 (def gen-tweet
2   (gen/hash-map :id      gen/large-integer
3                 :user-id gen/large-integer
4                 :text     gen/string
5                 :at       gen/large-integer))
6
7 (defspec tweet-roundtrip 100
8   (prop/for-all [tweet gen-tweet]
9     (do
10      (write-tweet-to-db tweet)
11      (= tweet (read-tweet-from-db (:id tweet)))))))
```

Getting Around

```
1  (require '[clojure.test.check.generators :as gen])
2
3  (gen/generator? gen/nat)
4  => true
5
6  (gen/sample gen/nat)
7  => (0 1 2 2 2 4 5 4 7 6)
8
9  (->> (gen/sample gen/nat 100) (reverse) (take 5))
10 => (65 56 19 57 13)
11
12 (gen/generate gen/nat)
13 => 5
```

Some Distinctions

```
1 (gen/generate gen/string-ascii) => "1|b$1cThVf"
2
3 (gen/generate (gen/vector gen/large-integer))
4 => [-95 -491 63140429 -3431 0 -311 -5
5     -2076 -45628915 -276774]
6
7 (def gen-vector-of-integers
8   (gen/vector gen/large-integer))
9
10 (gen/generate gen-vector-of-integers)
11 => [180584 -2 -22045 54270866 -437576 -11216]
```

Generating Simple Values

```
1 (gen/generate (gen/return 42)) => 42
2 (gen/generate gen/boolean) => true
3 (gen/generate gen/uuid)
4 => #uuid "bd7b1d9f-7d10-46f0-bf60-e7655bc30013"
5 (gen/generate (gen/elements [:foo :bar :baz]))
6 => :baz
7 (gen/generate (gen/shuffle [:a :b :c :d]))
8 => [:c :a :b :d]
9 (gen/generate gen/any-printable 5)
10 => [3/4 4 "" V2m!j-]
11 (gen/generate gen/simple-type-printable)
12 => 19/4
```

Generating Simple Values — Numbers & Strings

```
1 (gen/generate gen/nat) => 8
2 (gen/generate gen/large-integer) => 5908552
3 (gen/generate gen/double) => -0.05810546875
4 (gen/generate gen/ratio) => 25/11
5 (gen/generate gen/byte) => 1
6
7 (gen/generate gen/string-ascii) => "L}bOK\"dHF["
8 (gen/generate gen/string-alphanumeric)
9 => "0XXVKNr6T9ZmI5"
10 (gen/generate gen/keyword)
11 => :H?2I!p7+:+1w_*-g-26P+BB!N-:_P!k89327V0y5L5?v6
12 (gen/generate gen/symbol) => s*
```


Generating Collections

```
1 (gen/generate (gen/vector gen/ratio))
2 => [1/2 8/11 -8/27 -1/5 -25/26 -4/3
3     -23/16 -26/11 1/5 1/3 -9/22 17/23
4     -2/3 17/6 -10/9 -25/27 2/11]
5
6 (gen/generate (gen/map gen/string-alphanumeric
7                        gen/large-integer))
8 => {"QlfwE28o59osPwD3FYjAAkX7UGoI5n" -3526839,
9     "KGYJh06cKI0lMh0" -640463,
10    "VYzJ" 2,
11    "0" -191059464,
12    "OvBETs59ge2Clz29pQj63LZM8fqX5" -6,
13    "wr5dAjY23P4qlavDV957UPf40PmEjh" -12788626,
14    "140211UNiNE0wg9HG75" -128}
```

Generating Collections — Distinct Elements

```
1 (gen/generate
2   (gen/vector-distinct-by #(last (str %))
3     gen/large-integer
4     {:num-elements 10
5      :max-tries 1000}))
6 [921893
7  -4163099
8  -91274732683771995
9  -2
10 -13413836
11 5060233600
12 -197
13 -19709728
14 13863511
15 44704]
```

Generating Collections — Heterogeneous

```
1 (gen/generate
2   (gen/hash-map :user-id    gen/large-integer
3                  :parent-id gen/large-integer
4                  :text      gen/string-ascii))
5
6 => {:user-id 4441
7     :parent-id 1155
8     :text "R1f^DTs!?-ST0;9q1I-.]0/#L}z"}
9
10 (gen/generate
11   (gen/tuple gen/boolean gen/double gen/string-ascii))
12
13 => [false -29.0625 ">\"]
```

Combinators — gen/let

```
1 (gen/sample
2   (gen/let [x gen/nat]
3     [x (inc x)]))
4
5 => ([0 1] [1 2] [1 2] [1 2] [0 1]
6     [0 1] [2 3] [1 2] [8 9] [2 3])
7
8 (gen/generate
9   (gen/let [s gen/string-ascii
10             bounds (gen/vector (gen/large-integer
11                                 {:min 0, :max (count s)})
12                                 2)]
13     (let [[start end] (sort bounds)]
14       [s (subs s start end)])))
15
16 => ["HgSz!u1>nkhyxL|,Q:+/zms=#]2" "L|,Q:+/z"]
```

Combinators — gen/one-of and gen/such-that

```
1 (gen/sample (gen/one-of [gen/nat
2                      gen/boolean
3                      (gen/return nil)]))
4
5 => (true nil 2 false 1 true nil true true 4)
6
7 (gen/sample
8   (gen/such-that #(not= 1 (count %))
9                 (gen/list gen/nat)))
10
11 => (() () () (2 3 2) (0 3) (1 2 2 4)
12      (5 4 1 1 6 5) (1 1) (7 1) (4 6 9 7 5 3 8 2))
```

Generators

Exercises

Exercises – 4clojure

- <http://4clojure.gfredericks.com>
- cheatsheet: <https://github.com/clojure/test.check>

Testing

Testing

Properties, defspec

Anatomy of a spec

```
1 (require
2   '[clojure.test.check.clojure-test :refer [defspect]]
3   '[clojure.test.check.generators :as gen]
4   '[clojure.test.check.properties :as prop])
5
6 (defspect numbers-work-pretty-good 1000
7   (prop/for-all [x gen/nat
8                   y gen/nat]
9     (= (+ x y)
10        (+ y x))))
```

Running a spec

```
1  ;; clojure.test will run these as normal
2  ;; tests, but they can also be called as
3  ;; functions
4
5  (numbers-work-pretty-good)
6  => {:num-tests 1000
7      :result true
8      :seed 1446498477711}
```

Failing tests

```
1  (defspec numbers-work-pretty-good 1000
2    (prop/for-all [x gen/nat
3                   y gen/nat]
4      (= (- x y)
5         (- y x))))
6
7  (numbers-work-pretty-good)
8  => {:fail [0 3],
9      :failing-size 3,
10     :num-tests 4,
11     :result false,
12     :seed 1446498603099,
13     :shrunk {:depth 2, :result false,
14              :smallest [0 1], :total-nodes-visited 5}}
```

Testing Shrinking

A Failing Test

```
1 (def gen-user
2   (gen/hash-map :name gen/string-alphanumeric
3                 :age gen/nat
4                 :comments (gen/vector
5                             gen/string-ascii)))
6
7 (defspec users-can't-be-42-years-old
8   (prop/for-all [user gen-user]
9     (not= 42 (:age user))))
```

Can users be 42 years old?

```
1 (users-can't-be-42-years-old)
2 =>
3 {:fail [{:age 42,
4           :comments ["f~Bz;cyd{IYT'"][u^g3Zb]bqp^20x'yXbQ+"
5                       "cx}:ZiX<hdQ;Dl(tL?>mG#f(K8rkuw'"
6                       "M~$"
7                       ;; ... 29 more strings omitted ...
8                       ],
9           :name "Q5JD69vn5ebT8I5Y4PLtS8hw"}],
10 :failing-size 42,
11 :num-tests 43,
12 :result false,
13 :seed 1446500137527,
14 :shrunk {:depth 56,
15           :result false,
16           :smallest [{:age 42, :comments [], :name ""}],
17           :total-nodes-visited 254}}
```

A More Specific Failing Test

```
1 (defspec users-with-at-least-two-comments-can't-be-42-years
2   (prop/for-all [user gen-user]
3     (not (and (= 42 (:age user))
4               (<= 2 (count (:comments user)))))))
```


Can users with at least two comments be 42 years old?

```
1 (users-with-at-least-two-comments-can't-be-42-years-old)
2
3 =>
4 {:fail [{:age 42,
5         :comments ["CE(*bQ>G\\RHwa]t_b_OR3wJi\"9GD_aPOC"
6                  "#CSGaoB!{56zzc2{-o\";3Z"
7                  ;; ... 38 more strings omitted ...
8                  ],
9         :name "aRwWqcRV36N97Qy9e8"}]},
10 :failing-size 60,
11 :num-tests 61,
12 :result false,
13 :seed 1446500367916,
14 :shrunk {:depth 90,
15         :result false,
16         :smallest [{:age 42, :name ""
17                   :comments ["" ""]}]},
18 :total-nodes-visited 598}}
```

The Process of Shrinking — 1

```
1 (defspec twenty-four-is-the-highest-number
2   (prop/for-all [xs (gen/vector gen/int)]
3     (every? #(<= % 24) xs)))
```

The Process of Shrinking — 2

```
1 ;; Testing [-18 10 20 12 25 -23]...fail
2 ;; Testing [10 20 12 25 -23]...fail
3 ;; Testing [-18 20 12 25 -23]...fail
4 ;; Testing [20 12 25 -23]...fail
5 ;; Testing [10 12 25 -23]...fail
6 ;; Testing [12 25 -23]...fail
7 ;; Testing [20 25 -23]...fail
8 ;; Testing [25 -23]...fail
9 ;; Testing [12 -23]...pass
10 ;; Testing [-23]...pass
11 ;; Testing [25]...fail
12 ;; Testing [0 -23]...pass
13 ;; Testing []...pass
14 ;; Testing [0]...pass
15 ;; Testing [13]...pass
16 ;; Testing [19]...pass
17 ;; Testing [22]...pass
18 ;; Testing [24]...pass
19 ;; {:fail [[-18 10 20 12 25 -23]], ...
20 ;; :shrunk {..., :smallest [[25]], :total-nodes-visited 12}}
```

Shrinking Pitfalls — 1

```
1 (def gen-vector-of-ints
2   (gen/let [length gen/nat]
3     (gen/vector gen/nat length)))
4
5 (defspec lists-don't-contain-42
6   (prop/for-all [xs gen-vector-of-ints]
7     (not-any? #{42} xs)))
```

Shrinking Pitfalls — 2

```
1 (lists-don't-contain-42)
2
3 => {:fail [[29 6 9 33 32 3 40 30 23 42 41 41
4           38 30 26 9 27 8 28 3 18 12 3 43 6 6]],
5      :failing-size 43,
6      :num-tests 44,
7      :result false,
8      :seed 1447603220629,
9      :shrunk {:depth 13,
10              :result false,
11              :smallest [[0 0 0 0 0 0 0 0 0 0 42 0]],
12              :total-nodes-visited 36}}
```

Testing

What sort of properties should I write?

- Roundtripping
 - Storage
 - Serialization
- Doing things in different orders (when order isn't supposed to matter): `(= (f (g x)) (g (f x)))`
- Idempotency: `(= (f x) (f (f x)))`
- Run your code to make sure it doesn't crash
 - Especially meaningful if you have runtime assertions
- For some programs verifying that the output is correct is much easier than computing it
- Test complex optimized code using simple unoptimized code (test that they do the same thing)
- When rewriting something, test that the old version and the new version do the same thing

Testing Exercises

Exercises – Testing some codebases

- Write some tests with some of the given codebases

Applied

Property-based testing in general

PRO

More test coverage
Expose assumptions
Design to avoid edge-cases
Discover bugs in tools/libs

CON

Test-writing is slower
Tests can be harder to understand
Test-running is slower
Doesn't work well at very large scales

Additional temporary drawbacks for `test.check` in particular

- Missing more advanced generators
 - time, unicode strings, ...
- Missing some usability features
 - Re-running failures is difficult
 - Can't customize shrinking process
 - No mechanism for storing regression examples
 - No "fast mode" for a quick run of a whole test suite

How do I use this to test my hairy business application?

Testing Difficult Things

- Write Libraries
- Model the External World
 - users, other systems, the clock
- Model Your Application
- Use Schemas & Other Assertions

Applied Exercises

Go home and write property-based tests.

- test.chuck: github.com/gfredericks/test.chuck
- stateful-check: github.com/czan/stateful-check
- freenode#test.check